Anybody can do Value at Risk: A Teaching Study using Parametric Computation and Monte Carlo Simulation

Yun Hsing Cheung[1], Robert Powell[1]

Abstract

The three main Value at Risk (VaR) methodologies are historical, parametric and Monte Carlo Simulation. Cheung & Powell (2012), using a step-by-step teaching study, showed how a nonparametric historical VaR model could be constructed using Excel, thus benefitting teachers and researchers by providing them with a readily useable teaching study and an inexpensive and flexible VaR modelling option. This article extends that work by demonstrating how parametric and Monte Carlo Simulation VaR models can also be constructed in Excel, thus providing a total Excel modelling package encompassing all three VaR methods.

Keywords: Value at risk, Parametric value at risk, Monte Carlo simulation, Financial modelling, Pseudo-random number generator.

JEL Classification: G17.

---

[1] Edith Cowan University, Perth, Australia. Corresponding author Robert Powell, email r.powell@ecu.edu.au

Introduction

Cheung and Powell (2012) showed the procedures of doing one-step ahead Value at Risk (VaR) in Microsoft Excel using the non-parametric historical method. This paper extends this prior research by calculating VaR using parametric and Monte Carlo simulation methods. In the parametric method, the asset returns are assumed to follow a known probability distribution whilst the Monte Carlo method assumes that asset returns are driven by a known stochastic process.

The major attraction of using a nonparametric approach, as argued by Cheung and Powell (2012), is avoiding the misspecification of probability density functions of risk factors in an era of frequent financial disturbance. If trading conditions are deemed to be normal then the VaR calculation can be simplified considerably if the distributions of the risk factors can be assumed to belong to certain parametric families, such as normal or gamma distribution. This leads to the use of the parametric method. Some researchers, especially those with a statistical background, may find the use of the parametric method to derive VaR rather restrictive and over-simplified, preferring instead that the probability distributions of the risk factors are derived empirically. This can be done by Monte Carlo simulation if the mechanisms of changes in the risk factors are known. In this paper, we assume that a stochastic process can model the mechanism of changes in asset returns, thus the asset returns are presented as a probability distribution rather than values. Moreover, we incorporate a self-contended pseudo-random number generator into our Monte Carlo simulation method, which as far as we know is a first in financial modelling using an Excel 2007 spreadsheet.

There are several studies which compare the relative merits of historical, parametric and Monte Carlo VaR approaches, for example Lechner & Ovaert (2010), Deepak & Ramanathan (2009), Jorion (2001), Pritsker (1997) and Stambaugh (1996). In general these studies find that there is no particular best method. Parametric methods are simple to implement and very useful when returns follow a normal distribution, but they are not appropriate when there is non-normality such as asymmetry or leptokurtosis. Monte Carlo has the advantage of increasing the number of observations but it can be time-consuming and computer-intensive to implement. The historical method accurately measures past returns but it can be a poor estimator of future returns if the market has shifted. Stambaugh (1996) notes that each method has strengths and weaknesses and that they should not be viewed as competing methods but as alternatives which might be appropriate in certain circumstances. Different approaches may be appropriate for different types of portfolio, different purposes and different levels of resources available to invest in the analysis.

To illustrate the use of the two methods described in this paper, we continue the Cheung and Powell (2012) teaching study. Four listed shares (Coca Cola, Bank of America, Boeing and Verizon Communication) from the New York Stock Exchange are used to demonstrate the calculation of VaR of a single asset and a portfolio. In the case of a single asset, an investor has an exposure of $1 million ($V$) worth of Coca Cola shares at time t (any trading day after 3 August 2010, which is the closing share price date in our sample). The risk factor is share price (p), risk horizon is one trading day, historical data series is 10 years of daily adjusted closing prices (from 4 August 2001 to 3 August 2010, a total of 2,513 observations), and the level of confidence (α) is 95%. The question of interest is: in 95 out of a 100 times, what would be the worst daily loss the investor could experience by holding $1 million Coca Cola shares? In the case of a portfolio (using the same historical period, number of observations, risk horizon and confidence level used for the single asset above), the investor extends his/her share portfolio exposure (*V*) to $5 million, comprising $1 million Coca Cola (20%), $1.5 million Bank of America (30%), $1.5 million Boeing (30%), and $1 million Verizon (20%). Again, we ask the question: in 95 out of a 100 times, what would be the worst daily loss the investor could experience by holding this $5 million portfolio?

This paper is organised as follows. The next section discusses the application of the parametric method to a single asset. The third section describes the workings of the Monte Carlo simulation method, again only applied to a single asset. The fourth section expands the two methods to calculate VaR for a portfolio of assets. The fifth section compares and discusses the results from the various methods. The last section is the conclusion.

Parametric Method: Single Asset

Using the parametric method, the researcher specifies a probability distribution that characterises the likely values of a risk factor. Bachelier (1900) used the central limit theorem to derive a normal distribution for share price movements in the Paris Stock Exchange, and discovered that successive changes in share prices are approximately normal. This normality assumption for asset returns has been in place since then. However, in the Black-Scholes (1973) model, share prices are assumed log-normally distributed, consistent with continuous compounding.

The crucial step in the parametric method is to obtain the mean and standard deviation of the normal distribution from the historical data series. Once these values are obtained, we can proceed to calculate the 5% VaR return by entering 5% in the first argument of the Excel function *NORMINV* (probability, mean, standard deviation). The 5% VaR value is then calculated by multiplying the exposure by (1 – the absolute value of the 5% VaR return). To plot the parametric VaR diagram, we construct a table with 80 bins for the calculation of the relative frequencies of the normal distribution. In Excel, the probability density function of a normal distribution is calculated by *NORMDIST* (x, mean, standard deviation, cumulative) where *x* is the x-coordinates showing the daily returns, mean and standard deviation are the parameters of the normal distribution, and cumulative = FALSE for the probability density function. The execution of this procedure is presented as a screenshot in Table 1.

Table 1
Individual Asset Parametric VaR

This screenshot shows the historical data series (called "cocadaily1" in Cells C7:C2519). For brevity we only show the first few returns. V = \$1 million (as shown in Cell G13), risk horizon is 1 day, n is 2,512, and confidence level (α) is 95% (Cell G8). We find that the daily mean return is -0.004% (Cell G6), standard deviation is 1.40% (Cell G7), 5% VaR return is -2.31% (Cell G10), and the 5% VaR value is -\$23,123.61 (Cell G12). For Excel functions applied to each cell in the spreadsheet, see Column I.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1													
2		**Coca Cola: 95% VaR by Parametric Method**											
3													
4		**Data**			**Calculation of 5% VaR value**								
5			Daily										
6		Obs	Returns		Daily mean return		-0.004%		Cell(G6)'s Formula: =AVERAGE(cocadaily1)				
7		1	-1.71%		Daily stn dev		1.40%		Cell(G7)'s Formula: =STDEVP(cocadaily1)				
8		2	1.31%		Confidence level		95%		Cell(G8)'s Value = 0.95				
9		3	-3.36%		5% VaR in decimal pt		5.00%		Cell(G9)'s Formula: =1-G8				
10		4	0.10%		5% VaR return		-2.31%		Cell(G10)'s Formula: =NORMINV(G9,G6,G7)				
11		5	1.13%		Amount of investment		$1,000,000		Cell(G11)'s Value = 1000000				
12		6	1.93%		5% VaR Value		$23,123.61		Cell(G12)'s Formula: =ABS(G11*G10)				
13		7	0.10%										
14		8	-1.52%		**Data for Charting**								
15		9	-1.95%										
16		10	-1.15%		Min daily return		-12.33%						
17		11	0.42%		Max daily return		8.11%						
18		12	0.10%		Range		20.45%						
19		13	-1.47%		No of daily obs		2,512						
20		14	-1.07%										
21		15	-1.62%					Relative					
22		16	-0.22%		0	*x*	*f(x)*	frequency					
23		17	-0.88%		1	-8.00%	0.0000	0.0000	Cell(G23)'s Formula: =NORMDIST(F23,$G$6,$G$7,FALSE)				
24		18	-2.23%		2	-7.70%	0.0000	0.0000	Cell(H23)'s Formula: =G23/$G$83				
25		19	-2.39%		3	-7.40%	0.0000	0.0000					
26		20	-0.35%		4	-7.10%	0.0001	0.0000					
27		21	0.00%		5	-6.80%	0.0002	0.0000					
28		22	1.61%		6	-6.50%	0.0006	0.0000					
29		23	-0.92%		7	-6.20%	0.0017	0.0000					
30		24	-0.23%		8	-5.90%	0.0042	0.0000					
31		25	0.00%		9	-5.60%	0.0100	0.0000					
32		26	-2.80%		10	-5.30%	0.0230	0.0001					
33		27	1.41%		11	-5.00%	0.0503	0.0002					
34		28	-3.21%		12	-4.70%	0.1052	0.0003					

Armed with the relative frequencies, we plot the parametric one-day VaR for Coca Cola shares in Figure 1.

Figure 1
Parametric One-day 5% VaR, Coca Cola

This shows the histogram of Coca Cola returns and the corresponding 5% VaR line using the parametric method. Data is contained in Cell F25:H84 of Table 1 where the x-coordinates representing the returns are listed in Cells F23:F83, the absolute frequencies in Cells G23:G83, and the resulting relative frequencies in Cells H23:H83. The insertion of the 5% VaR return line is thoroughly discussed in Cheung and Powell (2012) and will not be repeated here.



Monte Carlo Simulation Method: Single Asset

Monte Carlo simulation relies heavily on probability theory to drive the simulation process. It involves conducting repeated trials of the values of the uncertain input(s) based on some known probability distribution(s) and some known process to produce a probability distribution for the output. That is, each uncertain input or parameter in the problem of interest is assumed to be a random variable with a known probability distribution. The output of the model, after a large number of trials or iterations, is also a probability distribution rather than a numerical value. In the context of VaR, the uncertain input is the one-step-ahead asset returns and the uncertain outputs are the 5% VaR return and value. The process linking the inputs with the output is the geometric Brownian motion process.

Intuitively, the researcher can think of simulation like scenario analysis. Instead of having three or five scenarios, the simulation process generates thousands or tens of thousands of scenarios. From this long list of scenarios, we gain a much better understanding of the nature of the problem, the most likely outcome and the extent of uncertainty surrounding it.

Instead of defining the probability distribution of the risk factor (in this case, the return of a share) as in the parametric method, the Monte Carlo simulation method derives the distribution of the share returns using a stochastic process. In most finance studies, we assume that asset prices, though largely unpredictable, follow a special type of stochastic process known as geometric Brownian motion, described by the following equation:

$$S_{t+\Delta t} = S_t \, e^{\left(k\,\Delta t + \sigma\,\varepsilon_t\,\sqrt{\Delta t}\right)} \tag{1}$$

where S_t is the share price at time t, e is the natural log, Δt is the time increment (expressed as portion of a year in terms of trading days, e.g. one trading day will yield $\Delta t =$ 1/251.4 of a trading year in our exercise), $k = \mu - \left(\sigma^2/2\right)$ is the expected return (which equals annualised mean return μ minus half of the annualised variance of return σ^2), and ε_t is the randomness at time t introduced to randomise the change in share price. The variable ε_t is a random number generated from a standard normal probability distribution, which has a mean of zero mean and a standard deviation of one. Sengupta (2004, pp.285-295) provides a solid discussion of equation (1).

The return of a share price can be obtained by rearranging equation (1) to yield equation (2):

$$R_{t+\Delta t} = \ln\left(\frac{S_{t+\Delta t}}{S_t}\right) = k\,\Delta t + \sigma\,\varepsilon_t\,\sqrt{\Delta t} \tag{2}$$

The key to our exercise is generating the future returns according to equation (2). The main problem in modelling and simulating stochastic processes is generating a stream of random numbers. Excel provides several ways to generate random numbers, some true ones and some pseudo ones. True random numbers between 0 and 1 can be generated by the Excel function *RAND ()*. The problem with true random numbers is their volatile nature, which means a new value is returned every time the worksheet is recalculated (e.g. by pressing F9). This can be problematic if the researcher wants to repeat the experiment with the same set of random numbers or to re-examine the simulation results. This is where pseudo-random numbers come into play. Pseudo-random numbers are generated by formulas. As long as the seed number is fixed, the set of random numbers will be fixed, which enable the researcher to have a second chance to re-examine the simulation results. Excel provides a pseudo-random number generator in its Random Number Generation tool in Data Analysis buried deep in the Data Ribbon. Figure 2 is a screenshot of Excel's Random Number Generator. The number of variables box is the number of random number columns desired by the researcher, and the number of random numbers box is the required number of rows. The seed number is any whole number selected by the researcher, which is fixed to a specific set random numbers. For example, every time the number 10 is re-entered, those same random numbers will be generated.

Figure 2
Excel's Random Number Generator Dialog Box



To avoid the tedious task of calling up and filling in the Random Number Generator dialog box every time the researcher wants to change the seed number or simulate another stream of pseudo-random numbers, we recommend that the researcher build their own pseudo-random number generator. This can be easily incorporated into the simulation model.

One of the most popular random number generators is the linear congruential method developed by Lehmer as discussed in Sheskin (2007, pp.402-407). Equation (3) is a multiplicative variant of the linear congruential method which is designed to generate a stream of uniformly distributed random numbers *x* between 0 and 1:

$$x_{i+1} = [(a\,x_i) \bmod m]/m \qquad (3)$$

where *mod* is the modulo operation (it is conducted in Excel by the function *MOD* (number, divisor)), $0 \le a$ is the multiplier (a recommended number for *a*, as used by most statisticians, is 7^5), *m* is the modulus and it has to be greater than *a* (a recommended number for *m,* as used by most statisticians, is 2^{31}-1 or 2^{31}), and lastly, $0 < x_0$ is the initial seed number or starting value. The longest possible length of non-degenerated and non-cycled random numbers of this method is the value of the modulus.

The random numbers (*x)* generated by equation (3) are uniformly distributed random numbers representing probabilities of the events that certain rates of return will occur. They have to be transformed into normally distributed numbers (ε) before incorporating into equation (2). The transformation is carried out the using the Excel function *NORMSINV* (probability) where the random numbers enter the function as the only argument.

If the researcher wishes to use true random numbers, the Excel calculation function needs to be set up before incorporating the *RAND ()* function (note that this Excel function does not have an argument). In Excel 2003 or before, go to Tools and then Options. Once the Options dialog box appears, go to the Calculation tab and tick the Iteration box and set Maximum iterations to 1 and Maximum change to 0.001 (see Figure 3). Once iteration is turned on, iterations are generated by pressing the F9 key (instead of the random numbers continually recalculating themselves). Excel then recalculates the worksheet the number of times specified in the Maximum iterations box (when you press the F9 key) or until the results between calculations change less than the amount specified in the Maximum change box.

Figure 3
Excel Options Dialog Box in Excel 2003



If using Excel 2007 (with Vista), click the Microsoft Office Button, then Excel Options at the bottom of the dialog box, select Excel Add-Ins, and then select Formulas on the left-hand side panel to display the dialog box below. In the Calculation Options section, tick the Enable Iteration Calculation box and set the Maximum iterations to 1 and Maximum change to 0.001 (see Figure 4).

Figure 4
Excel Options Dialog Box in Excel 2007



The above discussion lays the groundwork for performing Monte Carlo simulations for calculating 5% VaR return and value for an individual asset.

The simulation process for 5% VaR returns and value includes five steps. Step one calculates the parameters in the geometric Brownian motion process. Step two generates uniformly distributed pseudo-random numbers between 0 and 1. Step three converts the uniformly distributed random numbers from step one to normally distributed random numbers between 0 and 1. Step four applies the normally distributed random numbers into the geometric Brownian motion process to yield the simulated asset returns. The final step calculates 5% VaR returns and 5% VaR value in a fashion similar to that discussed in the parametric method. Table 2 succinctly captures the calculation of the Monte Carlo simulation process.

Table 2
Individual Asset Monte Carlo Simulation VaR

Cells C7:C2519 show our historical Daily return series, again called “cocadaily1”. Cells H6:H16 show the preliminary calculation of the parameters for the geometric Brownian motion process. The share price in Cell H9 is the closing price for Coca Cola on the last day of our data sample (3 August 2010). There are three parameters in equation (2) that are required to be calculated before any simulation can take place. They are the time increment (denoted by Δt in equation (2) but called “deltaT” in the simulation of returns), expected return (denoted by k in equation (2) and in the simulation of returns), and annualised standard deviation of the historical returns (denoted by σ in equation (2) but called “stndev” in the simulation of returns). Note that time increments are specified in relation to one year. The average annual trading days over the 10 years equals 251.4 days (Cell H10), therefore the time increment applied is 0.0040 (Cell H11). The remaining two parameters are calculated in Cells H16 and H15, and their respective values are -3.55% and 22.25%. Once the essential parameters of the geometric Brownian process are computed, we move to step two of the process, which generates 2,000 uniformly distributed pseudo-random numbers between 0 and 1. This involves executing equation (3) in Excel. The generation of the 2,000 pseudo-random numbers is performed in Cells E23:E2023 in our worksheet; for illustration purposes the table shows only the first seven random numbers generated after the initial seed number. The initial seed number we used is 230 (first entered in Cell H19 and then fed into the generation process via Cell E23). Using the recommended values for the multiplier and modulus, and the initial seed number of 230, Cells E24:E30 show the subsequent seed numbers while Cells F24:F30 show the seven pseudo-random numbers generated. The 2,000 random numbers then need to be converted into 2,000 standard normally distributed random numbers before they are used to simulate the 2,000 possible returns for the next trading day. The conversion is carried out by using the Excel function *NORMSINV* (probability), with the uniformly pseudo-random numbers entered as probabilities. The outcome is shown in Cells G24:G30. The final step in the simulation process is feeding the normally distributed pseudo-random numbers into the geometric Brownian motion process, where equation (2) is employed. The seven simulated possible returns for the next trading day are presented in Cells H24:H30.

	A	B	C	D	E	F	G	H	I	J	K
1											
2		**Coca Cola: 95% VaR by Monte Carlo Simulation Method**									
3											
4		**Data**			**Geometric Brownian Motion**						
5			Daily								
6		Obs	Returns		Numer of obs			2,512	Cell(H6)'s Formula: =COUNT(cocadaily1)		
7		1	-1.71%		Min daily return			-12.33%	Cell(H7)'s Formula: =MIN(cocadaily1)		
8		2	1.31%		Max daily return			8.11%	Cell(H8)'s Formula: =MAX(cocadaily1)		
9		3	-3.36%		Share price now (S_0)			56.27	Cell(H9)'s Formula: =data1!D2522		
10		4	0.10%		Number of trading days per yr			251.4	Cell(H10)'s Value = 251.4		
11		5	1.13%		Time increment (Δt) for 1 day			0.0040	Cell(H11)'s Formula: =1/H10		
12		6	1.93%		Average daily return			-0.004%	Cell(H12)'s Formula: =AVERAGE(cocadaily1)		
13		7	0.10%		Daily standard deviation			1.403%	Cell(H13)'s Formula: =STDEVP(cocadaily1)		
14		8	-1.52%		Annualised mean return for 1 year (μ)			-1.07%	Cell(H14)'s Formula: =H12*H10		
15		9	-1.95%		Annualised stn dev (σ)			22.25%	Cell(H15)'s Formula: =H13*SQRT(H10)		
16		10	-1.15%		Expected return (k)			-3.55%	Cell(H16)'s Formula: =H14-((H15^2)/2)		
17		11	0.42%								
18		12	0.10%		No of iterations or trials			2,000	Cell(H18)'s Value = 2000		
19		13	-1.47%		Seed			230	Cell(H19)'s Value = 230		
20		14	-1.07%		modulus (m)			2,147,483,647	Cell(H20)'s Formula: =2^31-1		
21		15	-1.62%								
22		16	-0.22%		Prelim #	RAND()	NORMSINV	Return			
23		17	-0.88%		230						
24		18	-2.23%		3865610	0.0018	-2.9112	-4.10%	Cell(E24)'s Formula: =MOD((7^5)*E23,m)		
25		19	-2.39%		544797860	0.2537	-0.6629	-0.94%	Cell(F24)'s Formula: =E24/m)		
26		20	-0.35%		1694845859	0.7892	0.8037	1.11%	Cell(G24)'s Formula: =NORMSINV(F24)		
27		21	0.00%		1051258405	0.4895	-0.0262	-0.05%	Cell(H24)'s Formula: =k*deltaT+stndev*G24*sqrt(deltaT)		
28		22	1.61%		1152048966	0.5365	0.0915	0.11%			
29		23	-0.92%		774410210	0.3606	-0.3568	-0.51%			
30		24	-0.23%		1761498650	0.8203	0.9164	1.27%			

Table 3
Monte Carlo One-day 5% VaR, Coca Cola

The 5% VaR return (Cell N8) is obtained by using the Excel function *SMALL* (array, k-th smallest value in the array). Note that “simreturn” in the formula in Cell N8 is the name given to the 2,000 daily simulated return series (H24:H2023) from Table 2.

	A	L	M	N	O	P	Q
1							
2							
3							
4		**Calculation of 95% VaR Value**					
5							
6		Confidence level		95%	Cell(N6)'s Value = 0.95		
7		Bottom 5% obs		99	Cell(N7)'s Formula: =(1-N6)*W11-1		
8		5% VaR		-2.318%	Cell(N8)'s Formula: =SMALL(simreturn,N7)		
9		Amount of investment		$1,000,000	Cell(N9)'s Value = 1000000		
10		5% VaR Value		$23,177.81	Cell(N10)'s Formula: =ABS(N9*N8)		

Table 3 shows the calculation of VaR return and VaR value. To plot the probability distribution for the simulated returns, we construct an 80-bin table from -8.00% to 8.00% (with bin size of 0.2%) and use the *FREQUENCY* (data array, bins array) function to calculate the number of returns that fall into each bin. We then use the relative frequencies to construct a scatter with a smooth line chart, as shown in Figure 5. Apparently the returns are not normally distributed, with the distribution skewed to the left and showing a jagged curve. As usual, a volatile 5% VaR return line is fitted to the diagram.

Figure 5
Monte Carlo One-day 5% VaR, Coca Cola



There are two important issues to consider in relation to the Monte Carlo simulation method. The first issue concerns the initial seed number. Since this can be any positive value, what is the appropriate number? The second issue relates to the number of iterations. In our example, we run 2,000 iterations, which is an *ad hoc* decision. Is there a minimum number of ideal iterations? In the following paragraphs, we briefly discuss these two issues.

In Table 2 the initial seed number used in the simulation process was arbitrarily selected as 230. The resultant 5% VaR return and the 5% VaR value are -2.318% and $23,177.81 respectively (Table 3). Table 4 uses a range of other seed numbers to illustrate that there is no appropriate initial seed number.

Table 4
Impact of Initial Seed Number on 5% VaR Return and Value

In addition to the initial seed number of 230 used in Table 2, we perform the same simulation process with another four initial seed numbers: 5; 1,520; 29,765; and 677,777 as shown in the first column with the resulting VaR returns and values shown in the ensuing columns.

Initial seed number	5% VaR return	5% VaR value
5	-2.40%	$23,972
230	-2.32%	$23,178
1,520	-2.28%	$22,832
29,765	-2.28%	$22,832
677,777	-2.41%	$24,099

The 5% VaR return, in this sample of five seed numbers, fluctuates between -2.41% to -2.28%, while the 5% VaR value fluctuates from $24,099 to $22,832. The differences in the latter are insignificant with respect to the exposure of $1 million. The differences will narrow as the number of iterations increases. In view of this, any initial seed number is acceptable as long as a large number of iterations are simulated as discussed in the ensuing paragraphs.

Put into context, the number of iterations n is the number of pseudo-random numbers (ε) we have to generate. The minimum number of trials n depends on how precise you want your simulation to be. Equation (4) gives the minimum number of iterations to achieve the desired accuracy D, defined as $D = y - \mu$ where y is the simulated value of the risk factor and μ is the mean of the probability distribution of the risk factor.

$$n = \left(\frac{z_{\alpha/2}\,\hat{\sigma}}{D} \right)^2 \tag{4}$$

Most researchers, however, ignore equation (4) and simulate at least 10,000 to 20,000 times, which should give an approximately normal distribution for the risk factor. In our teaching study, we simulate only 2,000 times for illustration purposes.

Parametric Method: Mutiple Asset Portfolio

Assume our investor increases their portfolio holdings by purchasing \$1.5 million shares in Bank of America (BoA). The investor now has a portfolio of \$2.5m with \$1m (40%) Coca Cola and \$1.5 million (60%) BoA. When additional assets are introduced into the portfolio, we need to account for correlation and covariance between the assets before calculating the VaR. We use the variance-covariance matrix, which is the approach used by RiskMetrics (J.P. Morgan & Reuters 1996), who introduced VaR. We start with a two asset portfolio. The steps involved are shown in Table 5, and further reading on this approach can be obtained in Choudhry (2004).

Table 5
Two Asset Parametric VaR

The table shows the calculation of VaR for a 2 asset portfolio (Coca Cola and BoA). Steps 1-4 are calculated individually for each of the 2 assets. Steps 5-9 calculate the portfolio standard deviation by first calculating portfolio mean, correlation coefficient, covariance and portfolio variance. Formulae are shown alongside each step. VaR is calculated based on the standard normal distribution as shown in steps 10-12. This process is based on similar examples by Choudhry (2004).

	A	B	C	D
1				
2		Coca Cola	BoA	
3	1. Obtain relative weightings (w)	40.00%	60.00%	
4	2. Calculate mean (μ) return for each asset	-0.004%	-0.022%	
5	3. Calculate stdev (σ) for each asset	1.403%	3.635%	
6	4. Calculate variance (σ^2) for each asset	0.0002	0.0013	
7	5. Calculate weighted portfolio mean return (μ_p)	-0.015%	Formula: =SUMPRODUCT($B$3:$C$3,$B$4:$C$4)	
8	6. Calculate correlation coefficient (ρ_{xy})	0.2941	Formula: =CORREL(cocadaily1,BoAdaily1)	
9	7. Calculate covariance ($\rho_{xy}\sigma_x\sigma_y$)	0.0002	Formula: =$B$8*$B$5*$C$5	
10	8. Calculate Portfolio variance (σ^2_p) for each asset	0.0006	Formula: =(B3^2*B6)+(C3^2*C6)+(2*B3*C3*B9)	
11	9. Portfolio stdev (σ_p) = square root $\sigma^2 p$	0.0241	Formula: =SQRT(B10)	
12	10. 5% VaR daily return	-3.97%	Formula: =NORMINV(0.05,$B$7,$B$11)	
13	11. Initial portfolio value	\$2,500,000		
14	12. 5% VaR value	-\$99,355.32	Formula: = B12*B13	

Matrix multiplication is required to calculate variance-covariance for several assets. Matrices need to be set up with the number of columns in matrix A equal to the number of rows in matrix B. To calculate the value of a matrix C from matrices A and B, (where i is the row index and j is the column index for matrix A, and j is the row index and k the column index for matrix B), the following formula is used:

$$C_{ik} = \sum_j A_{ij} B_{jk} \tag{5}$$

Let us assume our investor's portfolio consists of $5 million. In addition to the shares mentioned in the previous section, the investor has $1.5 million shares in Boeing and $1 million in Verizon. The portfolio now contains shares in four companies with 20% Coca Cola ($1 million), 30% BoA ($1.5 million), 30% Boeing ($1.5 million) and 20% Verizon ($1 million).

Variance and correlation matrices need to be created and multiplied together to form a variance-correlation matrix. This in turn multiplies with the variance matrix to create a variance-covariance matrix, which is then multiplied with the weightings to form a weighted variance-covariance matrix, the sum of which gives the portfolio standard deviation from which the VaR can be calculated as shown in Table 6.

Table 6

Multiple Asset Parametric VaR

	A	B	C	D	E	F	G
1		Coca Cola	BoA	Boeing	Verizon		**Formulae**
2							
3	**Variance Matrix**						
4		1.40%	3.64%	2.10%	1.86%		
5	COCA COLA	0.0140	0.0000	0.0000	0.0000		Cell $B$5 = B$4
6	BoA	0.0000	0.0364	0.0000	0.0000		Where row 4 is the daily standard deviation
7	BOEING	0.0000	0.0000	0.0210	0.0000		Copy formula to all relevant cells in matrix per LHS example.
8	VERIZON	0.0000	0.0000	0.0000	0.0186		
9							
10							
11	**Correlation Matrix**						
12	COCA COLA	1.0000	0.2941	0.3249	0.4032		Cell $B$12 = CORREL(cocadaily1,cocadaily1)
13	BoA	0.2941	1.0000	0.3715	0.3745		Cell $B$13 = CORREL(cocadaily1,BoAdaily1)
14	BOEING	0.3249	0.3715	1.0000	0.3550		Cell $B$14 = CORREL(cocadaily1,boeingdaily1)
15	VERIZON	0.4032	0.3745	0.3550	1.0000		Cell $B$15 = CORREL(cocadaily1,verizondaily1)
16							Copy formulae across, varying according to column
17							(e.g. Column C has BOAdaily as the first item in brackets).
18	**Variance-Correlation Matrix**						
19	COCA COLA	0.0140	0.0041	0.0046	0.0057		Cell $B$19 = MMULT($B5:$E5,B$12:B$15)
20	BoA	0.0107	0.0364	0.0135	0.0136		Copy formula to all cells in matrix.
21	BOEING	0.0068	0.0078	0.0210	0.0074		
22	VERIZON	0.0075	0.0070	0.0066	0.0186		
23							
24							
25	**Weighted Variance-Covariance Matrix**						
26		20.00%	30.00%	30.00%	20.00%		
27	COCA COLA	0.0000	0.0000	0.0000	0.0000	20.00%	Cell $B$27 = MMULT($B19:$E19,B$5:B$8)* B$26* $F27
28	BoA	0.0000	0.0001	0.0000	0.0000	30.00%	Where row 26 and colum F are the weightings
29	BOEING	0.0000	0.0000	0.0000	0.0000	30.00%	Copy formula to all cells in matrix.
30	VERIZON	0.0000	0.0000	0.0000	0.0000	20.00%	
31							
32		Portfolio mean return			-0.0063%		
33		Portfolio Variance			0.03%		Cell $E$33 = SUM(B27:E30)
34		Standard Deviation			1.78%		Cell $E$34 = SQRT(E33)
35		5% VaR			-2.93%		Cell $B$35 = Norminv(0.05,E32,E34)
36		Amount of Investment			$5,000,000		
37		5% VaR Value			-$ 146,507		Cell $E$37 = E36* E35

The table shows matrix multiplication for the four share portfolio. The historical return series for each of the four assets are named cocadaily1, BoAdaily1, boeingdaily1 and verizondaily1. Further shares can be accommodated by increasing the number of rows and columns equally, limited only by the number of columns in Excel. Matrices in Excel can be multiplied together using the formula *MMULT ()* as shown in the formulae in Column G. The variance matrix is multiplied by the correlation matrix to form the variance-correlation matrix, which is then multiplied by the variance matrix and share weightings to form the variance-covariance matrix. The latter is summed to calculate the portfolio variance from which the standard deviation and VaR are calculated as per Rows 32:37. It should be noted that, if preferred, the Excel Data Analysis Add-in can be used an alternative tool to generate individual matrices such as the correlation matrix.

Monte Carlo Simulation Method: Multiple Asset Portfolio

First undertake a Monte Carlo simulation for each asset in the portfolio. Then obtain the daily weighted average returns from which VaR is calculated as per Table 7.

Table 7

Multiple Asset Monte Carlo Simulation VaR

	A	B	C	D	E	F	G
1							
2	**Historical Returns:**						
3			Coca Cola	BoA	Boeing	Verizon	
4		1	-1.71%	4.00%	-0.13%	0.40%	
5		2	1.31%	-0.96%	-0.64%	-5.02%	
6		3	-3.36%	-0.24%	0.00%	-6.03%	
7		4	0.10%	2.38%	-2.21%	-4.54%	
8		5	1.13%	0.35%	1.95%	2.45%	
9		6	1.93%	0.23%	2.29%	-0.91%	
10		7	0.10%	2.42%	-2.16%	1.36%	
11		8	-1.52%	-0.80%	-3.54%	0.45%	
12		9	-1.95%	-1.51%	-2.16%	-1.36%	
13		10	-1.15%	-0.47%	0.14%	-2.77%	
14		11	0.42%	0.82%	0.00%	8.23%	
15		12	0.10%	0.23%	3.87%	-1.01%	

	H	I	J	K	L
3		Coca Cola	BoA	Boeing	Verizon
4	Numer of obs	2,512	2,512	2,512	2,512
5	Min daily return	-12.33%	-37.64%	-23.07%	-14.70%
6	Max daily return	8.11%	31.52%	12.31%	12.32%
7	Share price now (S_0)	56.27	14.36	37.60	29.64
8	Number of trading days per yr	251.4	251.4	251.4	251.4
9	Time increment (Δt) for 1 day	0.0040	0.0040	0.0040	0.0040
10	Average daily return	0.00%	-0.02%	0.01%	-0.01%
11	Daily standard deviation	1.40%	3.64%	2.10%	1.86%
12	Annualised mean return (μ)	-1.07%	-5.65%	3.47%	-3.52%
13	Annualised stn dev (σ)	22.25%	57.65%	33.25%	29.47%
14	Expected return (k)	-3.55%	-22.27%	-2.06%	-7.86%
16	No of iterations or trials	2,000	2,000	2,000	2,000
17	Seed	230	500	750	1,000
18	modulus (m)	2,147,483,647	2,147,483,647	2,147,483,647	2,147,483,647

	A	B	C	D	E	F
19						
20						
21	**Simulated Returns:**					
22		Weighted average	Coca Cola	BoA	Boeing	Verizon
23		Weighting:	20%	30%	30%	20%
24	1	0.861%	-0.944%	2.584%	0.819%	0.145%
25	2	1.552%	1.114%	3.006%	1.115%	0.465%
26	3	1.015%	-0.051%	-1.738%	4.697%	0.685%
27	4	0.056%	0.114%	-1.165%	0.391%	1.325%
28	5	-0.034%	-0.515%	-0.275%	-1.631%	3.204%
29	6	-0.712%	1.272%	0.115%	-1.209%	-3.189%
30	7	-0.775%	-1.537%	-0.708%	-2.194%	2.016%
31	8	-0.065%	-1.013%	1.295%	-0.154%	-1.025%
32	9	-1.005%	2.850%	-1.524%	-4.328%	0.905%
33	10	0.675%	-1.168%	2.375%	0.668%	-0.022%
34	11	0.556%	-0.135%	1.904%	0.320%	-0.424%
35	12	-0.215%	-1.649%	-0.246%	-1.599%	3.344%

	H	I	J
23	**Calculation of 95% VaR Value**		
25	Confidence level	95%	
26	Bottom 5% obs	99	
27	5% VaR	-2.71%	Cell(I27)'s formula: = SMALL(weightedsimreturn,I25)
28	Amount of investment	$5,000,000	
29	5% VaR Value	$135,307.57	

An identical simulation process is followed for each of the four shares in our portfolio as was followed for Coca Cola in Table 2, and the daily weighted average returns are then calculated. The summarised results are shown in Cells B24:F35. VaR is then calculated in Cells I25:I29 for the weighted average returns in exactly the same manner as was used for a single asset in Table 3. Formulae are not repeated from Tables 2 and 3. The share prices in Cells I7:L7 are the closing prices of the last day of our data sample (3 August 2010).

A Comparison of the Teaching Studies Results

Table 8
Comparison of Results from Various VaR Methods

The 5% VaR returns and values calculated from the various methods are shown in the table. Historical and Historical bootstrap results are extracted from Cheung and Powell (2012) who use identical data to this study. Parametric and Monte Carlo results are obtained from this study (individual asset results from Tables 1 and 3 with multiple assets results from Tables 6 and 8).

	Individual Asset (Coca Cola)		**Portfolio of Multiple Assets**	
Method	5% VaR return	5% VaR value	5% VaR return	5% VaR value
Historical	-2.20%	$21,979	-2.63%	$131,334
Historical bootstrap	-2.20%	$21,978	-2.63%	$131,334
Parametric	-2.31%	$23,124	-2.93%	$146,507
Monte Carlo simulation	-2.32%	$23,178	-2.71%	$135,308

The smallest and the largest 5% VaR returns in Table 8 differ by 0.12% (Coca Cola) and 0.30% (portfolio), while the smallest and the largest 5% VaR values differ by $1,200 (Coca Cola) and $15,163 (multiple asset portfolio). These differences are insignificant given the portfolio sizes of $1 million (individual asset) and $5 million (multiple asset). Based on the similar results obtained, it is difficult to argue which method is better. Indeed, the results depend on the method and the historical data series collected. Further back testing, beyond the scope of this paper, needs to be performed to yield further information to ascertain the appropriateness of these methods (Berry 2009).

Conclusion

The study, together with the prior work of Cheung and Powell (2012), shows how a complete range of VaR models, encompassing all three main VaR methods, can be constructed in Excel. The step-by-step teaching study approach allows teachers, students and researchers to build inexpensive VaR models. These range from simplistic parametric methods suitable for normal trading conditions through to more complex historical and (most complex) Monte Carlo models not dependent on a normal distribution assumption and more suited in times of frequent financial disturbance. The Excel models are highly flexible and easy to change as well as offering a range of modelling techniques such as the real or pseudo random number generators.

References

Bachelier, L. 1900, *Théorie de la Spéculation*. Paris: Gauthier-Villars. Translated by A. J. Boness. In *The Random Character of Stock Market Prices*, ed. P. Cootner 1964, pp.17-78. Cambridge, MA: The MIT Press.

Berry, R. 2009, *Back Testing Value-at-Risk.* Retrieved 1 September, 2011. Available at http://www.jpmorgan.com/tss/General/Back_Testing_Value-at-Risk/1159398587967

Black, F. & Scholes, M. 1973, 'The Pricing of Options and Corporate Liabilities', *Journal of Political Economy,* vol.8, no.13, pp.637-654.

Cheung, Y. H. & Powell, R. J. 2012, 'Anybody Can Do Value at Risk: A Nonparametric Teaching Study', *Australasian Accounting, Business and Finance Journal*, vol.6, no.1, pp.111-123.

Choudhry, M. 2004, *Fixed Income Markets, Instruments, Applications, Mathematics,* John Wiley & Sons (Asia) Pty Ltd, Singapore.

Deepak, J. & Ramanathan, T. 2009, 'Parametric and Nonparametric Estimation of Value at Risk'. *The Journal of Risk Model Validation,* vol.3, no.1, pp.51-71.

Jorion, P. 2001, *Value at Risk: The New Benchmark for Managing Financial Risk*. 2nd edn, McGraw-Hill, New York.

J.P. Morgan & Reuters 1996, *RiskMetrics Technical Document*. Retrieved 14 August, 2011. Available at www.faculty.idc.ac.il/kobi/RiskMGT/rmtd.pdf

Lechner, A. & Ovaert, T. 2010, 'Techniques to Account for Leptokurtosis and Assymetric Behaviour in Returns Distributions', *Journal of Risk Finance,* vol.11, no.5, pp.464-480.

Pritsker, M. 1997. 'Evaluating Value-at-Risk Methodologies: Accuracy versus Computational Time', *Journal of Financial Services Research,* vol.12, no.2/3, pp.201-242.

Sengupta, C. 2004, *Financial Modeling Using Excel and VBA*. John Wiley & Sons, New York.

Sheskin, D. J. 2007, *Handbook of Parametric and Nonparametric Statistical Procedures*. 4th edn, Chapman & Hall/CRC, Boca Raton, FL.

Stambaugh, F. 1996, 'Risk and Value at Risk', *European Management Journal*, vol.14, no.6, pp.612-621.